

Visualizing Free and Forced Harmonic Oscillations using Maple V R5*

HARALD PLEYM

Telemark College, Department of Technology, Kjolnes Ring 56, 3914 Porsgrunn, Norway.

E-mail: harald.pleym@hit.no

This paper focuses on how the symbolic, numerical and graphical power of the computer algebra system Maple V Release 5 can be used to explore and visualize with animation free harmonic motion with and without damping and vibration of a mass-spring system forced by an external sharp blow. Dirac's delta function is used to model the external force that acts for a very short period of time. It is also demonstrated how displacement resonance in a forced oscillation can be readily obtained from the transfer function of the system represented by a second-order differential equation. The information contained in the system frequency response may also be conveniently displayed in animated graphical form.

INTRODUCTION

AT TELEMARKE College, Department of Technology in Porsgrunn, Norway we have incorporated the use of the computer algebra system Maple V in the engineering mathematics curriculum from August 1995. With the powerful software program including graphical, symbolic and numerical techniques, Maple V Release 5 is the ideal program to make mathematics more relevant and motivating for engineering students and an excellent demonstration tool for the classroom. The use of Maple has changed the way I teach and hopefully the way the students learn. I now have the opportunity to emphasize the learning of concepts, the visualization of concepts and the solution of more realistic engineering applications.

Many of the phenomena we observe in everyday life have a periodic motion. And in many cases the periodic motion is maintained by a periodic driving force. For these forced oscillations the amplitude depends on the frequency of the driving force. There is usually some frequency of the driving force at which the oscillations have their maximum amplitude. This occurs when the natural frequency of the vibrating system and the frequency of the driving force are approximately equal. This phenomenon of resonance plays an important role in almost every branch of physics. And the avoidance of destructive resonance vibrations is an important factor in the design of mechanical systems of all types.

Application of differential equations plays an important role in science and engineering. And the most important step in determining the natural frequency of vibration of a system is the formulation of its differential equation. The purpose of this article is to demonstrate how Maple can be used to investigate and visualize with animations:

- simple harmonic motion of a mass-spring system;
- the response of a mechanical system (mass-spring system) to an impulse function;
- the displacement resonance in forced oscillations.

ANIMATION

Coding in Maple does not require expert programming skills, so writing a Maple program deals with putting a `proc()` and an end around a sequence of commands. And it is pretty easy to put pre-written routines together from Maple's powerful building blocks and plotting facilities. The `plottools` package provides many useful commands for producing plotting objects, which can be scaled, translated and rotated. The following Maple codes will be used in the coming sections. The first procedure animates a mass-spring system with and without a dashpot, and the second animates the response to a rectangular pulse by a linear differential second-order equation.

Because it is impossible to show animation on paper, the animation figures show four consecutive frames. The animations in this paper can be seen live on the Journal's Web site.

* Accepted 28 June 1999.

Maple code for a mass-spring system

Given $m = \text{mass}$; $r = \text{damping constant}$; $k = \text{spring constant}$; $x_1, x_2 = \text{positions of the spring}$ $x_0 = x(0)$, $x_{p0} = D(x)(0) = \text{initial conditions}$; $f = \text{driving force}$; $n = \text{number of frames}$.

```
>mass_spring:=proc(m,r,k,x1,x2,x0,xp0,tk,f,n)
  local
mass,deq,init,sol,xk,xu,plt1,plt2,plt,pltxk,rect,base,spring,spring1,rod,cylinder,piston,fluid,dashpot;
  with(plots):with(plottools): #loads the plots and plottools packages
  spring:=proc(x1,x2,n) #procedure for the spring
    local p1,p2,p3,p4,pn_1,pn,p;
    p1:=[x1,0];p2:=[x1+0.25,0];p3:=[x2-0.25,0];p4:=[x2,0];
    p:=i->[x1+0.25+(x2-x1-0.5)/n*i,(-1)^(i+1)*0.5];
    plot([p1,p2,seq(p(i),i=1..n-1),p3,p4],thickness=2,color=aquamarine);
  end:
spring1:=x2->spring(x1,x2,12);
rect:=t->translate(rectangle([-0.2,-0.2],[0.2,0.2],color=red),t,xk(t)+x2+1);
#rect: display the position of the mass on the displacement curve x(t)
mass:=x2->rectangle([x2,-1],[x2+2,1],color=red):
if x0=0 and xk(0.1)=0 then xu:=1.5
  elif x0=0 and xk(0.1)<>0 then xu:=3.5;
  else xu:=0;
fi;
rod:=x2->plot([[x2+2,0],[x2+2.5+2*xu,0]],color=grey,thickness=2);
cylinder:=plot([[x2+2.5xu,-1],[x2+2.5+xu,1],[x2+3.5+3*xu,1],[x2+3.5+3*xu,-1],[x2+2.5+xu,-1]],color=tan,thickness=2);
piston:=x2->rectangle([x2+2.5+2*xu,-1],[x2+3+2*xu,1],color=grey);
fluid:=rectangle([x2+2.5+xu,-1],[x2+3.5+3*xu,1],color=blue);
dashpot:=x2->display(rod(x2),cylinder,piston(x2),fluid);
deq:=m*diff(x(t),t$2)+r*diff(x(t),t)+k*x(t)=f(t);
print(deq);
init:=x(0)=x0,D(x)(0)=xp0;
sol:=dsolve({deq,init},x(t));
print(combine(simplify(sol)));
xk:=unapply(rhs(sol),t);
#xk=x(t) : the displacement of mass
pltxk:=plot([xk(t)+x2+1,x2+1],t=0..max(tk,x2+4+3*xu)+0.5,
color=[blue,greyl],numpoints=400);
base:=plot(-1.1,t=0..max(tk,x2+4+3*xu)+0.5,color=brown,thickness=4);
if r=0 then #undamped case
  plt1:=x->translate(display(spring1(x2+x),mass(x2+x),base),0,-3);
  else #with damping
plt1:=x->translate(display(spring1(x2+x),mass(x2+x),dashpot(x2+x),base),0,-3);
  fi;
  plt2:=i->display(pltxk,rect(tk/n*i));
  plt:=i->display(plt1(xk(tk/n*i)),plt2(i));
  display(seq(plt(i),i=0..n),insequence=true);
  display(% ,args[11..nargs]);
end:
```

Maplecode for a rectangular pulse

Given $m = \text{mass}$; $r = \text{damping constant}$; $k = \text{spring constant}$; $a = \text{constant}$; $\text{eps} = \text{constant}$; $n = \text{number of frames}$.

```
>impulse_func:=proc(m,r,k,a,eps,n)
  local h,dlign1,dlign2,sol1,sol2,plt2,pltd,plt,txt,txtd,kloss,e;
  with(plots):
  alias(u=Heaviside):
  h:=eps->(u(t-a)-u(t-a-eps))/eps;
  dlign1:=eps->m*diff(y(t),t,t)+r*diff(y(t),t)+k*y(t)=5*h(eps);
  dlign2:=m*diff(y(t),t,t)+r*diff(y(t),t)+k*y(t)=5*Dirac(t-a);
  print(dlign1(eps),epsilon=eps);
  print(dlign2);
  sol1:=eps->simplify(dsolve({dsolve({dlign1(eps),y(0)=0,D(y)(0)=0},y(t))});
  sol2:=dsolve({dlign2,y(0)=0,D(y)(0)=0},y(t));
  e:=textplot([6,1.3,convert([101],bytes)],font=[SYMBOL,12]):
  txt:=eps->textplot([6.9,1.3,cat('=',convert(evalf(eps,2),string))]);
  align={RIGHT,ABOVE};
  pltd:=eps->plot(h(eps),t=0..5*Pi,color=green,thickness=2):
```

```

plt1:=eps->plot(rhs(sol1(eps)),t=0..5*Pi,color=red,thickness=2):
plt2:=plot(rhs(sol2),t=0..5*Pi,color=blue,thickness=2):
plt:=eps->display(plt1(eps),plt2(eps),e,txt(eps),e,txt(eps),plt1(eps),plt2);
display(seq(plt(eps-(eps-0.1)/n*i),i=0..n),insequence=true);
display(% ,args[7..nargs]);
end:

```

FREE HARMONIC OSCILLATIONS

Machines with rotating components commonly involve mass-spring systems or their equivalents in which the driving force is simple harmonic motion. The motion of a mass attached to a spring serves as a simple example of vibrations that occur in more complex mechanical systems.

From a teaching point of view it is suitable to consider a body of mass m attached to one end of a spring that resists compression as well as stretching. A rod attached to the mass carries a disk moving in an oil-filled cylinder (a dashpot). The other end of the spring could be attached to a fixed wall and vibrating horizontally. The resultant force on the body is the sum of the restoring force $-kx(t)$ and the damping force $-r(\partial/\partial t)x(t)$ where k is the force constant, $x(t)$ is the distance of the body of mass from its equilibrium position, t is time and r is the damping constant. We take $x(t) > 0$ when the spring is stretched. The differential equation of motion is therefore:

```
>deq1:=m*diff(x(t),t$2)+r*diff(x(t),t)+k*x(t)=0;
```

$$deq1 := m \left(\frac{\partial^2}{\partial t^2} x(t) \right) + r \left(\frac{\partial}{\partial t} x(t) \right) + kx(t) = 0$$

If we set $r = 0$ in $deq1$ the motion is undamped. Otherwise the solution of $deq1$ presents three distinct cases of damping according to whether $d = r^2 - 4mk$ is greater than, equal to, or less than zero.

```
>eq:=d=r^2-4*m*k;
>solm:=solve(eq,m);
```

Let us replace m with a new variable mv in $deq1$.

```
>mv:=unapply(solm,d,r,k);
```

$$mv := (d, r, k) \rightarrow \frac{1-d+r^2}{4k}$$

```
>deq2:=collect(4*k*subs(m=mv(d,r,k),deq1),diff(x(t),t$2));
```

$$deq2 := (-d + r^2) \left(\frac{\partial^2}{\partial t^2} x(t) \right) + 4k \left(r \left(\frac{\partial}{\partial t} x(t) \right) + kx(t) \right) = 0$$

With this differential equation, $deq2$, the system is overdamped if $d > 0$, critically damped if $d = 0$ and underdamped if $d < 0$. If we request Maple to solve $deq1$ for the undamped case and $deq2$ for each of the damped cases, subject to initial condition $x(0) = x_0$ and $((\partial/\partial t)x(t))|_{t=0} = 0$.

```
>init:=x(0)=x[0],D(x)(0)=0: #initial conditions
```

we get the following solutions.

Undamped conditions

With $r = 0(Ns/m)$ we get:

```
>deq1a:=subs(r=0,deq1);
```

$$deq1a := m \left(\frac{\partial^2}{\partial t^2} x(t) \right) + kx(t) = 0$$

```
>dsolve({deq1a,init},x(t));
```

$$x(t) = x_0 \cos \left(\sqrt{\frac{k}{m}} t \right)$$

A typical graph of $x(t)$ with $m = 1 \text{ kg}$ and $k = 1 (N/m)$ is shown in Fig. 1. Figure 2 shows the animation of the undamped motion

```
>xk:unapply(rhs(%),m,k,x[0],t): #defines x=x(m,k,x[0],t)
```

Overdamped conditions

```
>interface(showassumed=0):
```

```
>assume(d>0):
```

```
>dsol:=dsolve({deq2,init},x(t));
```

$$dsol := x(t) = \frac{1}{2} \frac{x_0(-r + \sqrt{d})e^{(-2(r+\sqrt{d})kt/-d+r^2)}}{\sqrt{d}} + \frac{1}{2} \frac{x_0(r + \sqrt{d})e^{2((-r+\sqrt{d})kt/-d+r^2)}}{\sqrt{d}}$$

```
>subs(eq,dsol);
```

$$x(t) = \frac{1}{2} \frac{x_0(-r + \sqrt{r^2 - 4mk})e^{(-1/2(r+\sqrt{r^2-4mk})t/m)}}{\sqrt{r^2 - 4mk}} + \frac{1}{2} \frac{x_0(r + \sqrt{r^2 - 4mk})e^{(1/2(-r+\sqrt{r^2-4mk})t/m)}}{\sqrt{r^2 - 4mk}}$$

```
>xo:=unapply(rhs(%),m,r,k,x[0],t): #defines x=x(m,r,k,x[0],t)
```

The solution consists of two exponential terms when $r > \sqrt{4mk}$. In all subsequent figures we take $m = 1 \text{ kg}$ and $k = 1 (N/m)$. Figure 3 show some typical graphs of the position function for the overdamped case. Figure 4 shows the animation of the motion with the damping constant $r = 4 (Ns/m)$.

Critically damped condition

```
>lhs(dsol)=limit(rhs(dsol),d=0);
```

$$x(t) = \frac{x_0(2kt + r)e^{(-2(kt/r))}}{r}$$

```
>subs(r=sqrt(4*m*k),%);
```

$$x(t) = \frac{1}{2} \frac{x_0(2kt + 2\sqrt{mk})e^{-(kt/\sqrt{mk})}}{\sqrt{mk}}$$

```
>xc:unapply(rhs(%),m,k,x[0],t): #defines x=x(m,k,x[0],t)
```

The solution consists of one exponential term when $r = \sqrt{4mk}$. The graph in Fig. 5 resemble those of the overdamped case in Fig. 3. The animation of a critically damped motion with $r = 2Ns/m$ is shown in Fig. 6.

Under-damped condition

```
>assume(d<0):
```

```
dsolve({deq2,init},x(t));
```

$$x(t) = \frac{x_0 r e^{(2rkt/d-r^2)} \sin\left(2 \frac{\sqrt{-d}kt}{d-r^2}\right)}{\sqrt{-d}} + x_0 e^{(2rkt/d-r^2)} \cos\left(2 \frac{\sqrt{-d}kt}{d-r^2}\right)$$

```
>subs(eq,%);
```

$$x(t) = \frac{x_0 r e^{(-1/2(rt/m))} \sin\left(-\frac{1}{2} \frac{\sqrt{-r^2 + 4mkt}}{m}\right)}{\sqrt{-r^2 + 4mk}} + x_0 e^{(-1/2(rt/m))} \cos\left(-\frac{1}{2} \frac{\sqrt{-r^2 + 4mkt}}{m}\right)$$

```
>xu:unapply(rhs(%),m,r,k,x[0],t): #defines x=x(m,r,k,x[0],t)
```

The solution represents exponentially damped oscillations of the mass-spring system about its equilibrium position as shown in both Fig. 7 and Fig. 8.

Undamped motion results

```
>with(plots): #load the plots package
```

```
>plot(xk(1,1,1,t),t=0..19,color=black,labels=['t','x(t)'],labelfont=[TIMES,BOLD,12]);
```




Fig. 1. Undamped motion. $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$.

```
>mass_spring(1,0,1,0,5,2,0,14,0,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + x(t) = 0$$

$$x(t) = 2 \cos(t)$$




Fig. 2. Animation of an undamped motion. $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$.

The small rectangle shows the position of the mass on the displacement curve.

Overdamped motion results

If we select $m = 1 \text{ kg}$, $k = 1(N/m)$ and three different d -values: 12, 45, 96 $((Ns/m)^2)$, r named rk becomes:

```
>rk:=(d,m,k)->simplify(sqrt(d+4*m*k)):
r:=seq(rk(d,1,1),d=[12,45,96]);
```

$$r = [4, 7, 10]$$

```
>plt1:=d->plot(xo(1,rk(d,1,1),1,2,t),t=0..20,color=black):
>ttx1:=d->textplot([4,xo(1,rk(d,1,1),1,2,4),convert(r=rk(d,1,1),string)]):
>pltod:=display(seq({plt1(d),ttx1(d)},d=[12,45,96]),labels=['t','x(t)'],labelfont=[TIMES,BOLD,12]);%
```




Fig. 3. Overdamped motion, $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, r : damping constant.

Figure 3 shows three typical graphs of the position function $x(t)$ for the overdamped case. Figure 4 shows the animation of the motion with $r = 4(Ns/m)$.

```
>mass_spring(1,4,1,0,5,2,0,14,0,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + 4\left(\frac{\partial}{\partial x}x(t)\right) = 0$$

$$x(t) = \frac{2}{3}e^{(-2+\sqrt{3})t}\sqrt{3} + e^{(-2+\sqrt{3})t} - \frac{2}{3}\sqrt{3}e^{(-2+\sqrt{3})t} + e^{(-2+\sqrt{3})t}$$




Fig. 4. Animation of an overdamped motion of a mass-spring system with dashpot: $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, $r = 4(Ns/m)$.

Critically damped motion results

```
>plt2:=plot(xc(1,1,2,t),t=0..20,color=black):
>txt2:=textplot([3,xc(1,1,2,3),convert(r=rk(0,1,1),string)]):
>pltcd:=display(plt2,txt2):%;
```




Fig. 5. Critically damped motion, $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, r : damping constant.

```
>mass_spring(1,2,1,0,5,2,0,14,0,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2} x(t)\right) + 2\left(\frac{\partial}{\partial t}\right)x(t) = 0$$

$$x(t) = 2e^{(-t)} + 2e^{(-t)}t$$




Fig. 6. Animation of an critically damped motion. $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, $r = 2(Ns/m)$.

In this critically damped case, then resistance of the dashpot is just large enough to damp out any oscillations.

Underdamped motion results

We select $m = 1 \text{ kg}$, $k = 1(N/m)$ and two r values $\frac{1}{2}$ and $\frac{1}{5}$ (Ns/m):

```
>plt3:=plot([xu(1,1/2,1,2,t),xu(1,1/5,1,2,t)],t=0..20,color=black):
>t3:=textplot([6.5,xu(1,1/2,1,2,6.5),'r=1/2'],[6.5,xu(1,1/5,1,2,6.5),'r=1/5']):
>pltod:=display(plt3,t3,labels=['t','x(t)'],labelfont=[TIMES,BOLD,12]):%
```




Fig. 7. Underdamped motion, $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, r : damping constant.

```
>mass_spring(1,1/5,1,0,5,2,0,20,0,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + \frac{1}{5}\left(\frac{\partial}{\partial t}x(t)\right) + x(t) = 0$$

$$x(t) = 2e^{(-1/10)t} \cos\left(\frac{3}{10}\sqrt{11}t\right) + \frac{2}{33}\sqrt{11}e^{(-1/10)t} \sin\left(\frac{3}{10}\sqrt{11}t\right)$$

Fig. 8. Animation of an underdamped motion of a mass-spring system with dashpot: $m = 1 \text{ kg}$, $k = 1(N/m)$, $r = (1/5)(Ns/m)$.

The action of the dashpot exponentially damps the oscillations in accord with the time-varying amplitude. The dashpot also decreases the frequency of the motion from 1 in the undamped to case $3\sqrt{11}/10$ in the underdamped motion with the same mass and force constant.

Comparison of the graphs in Figs 3, 5 and 7 (Fig. 9) shows that when the motion is critically damped the mass reaches its equilibrium position in a shorter time than when the damping is larger. And for any damping constant less than $r = 2(Nm/m)$ in our particular example the motion becomes oscillatory.

```
>display(plt1(45),plt2,plt3,t3,labels=['t','x(t)'],labelfont=[TIMES,BOLD,12]);
```





Fig. 8.




Fig. 9. Overdamped, critically damped and damped oscillatory motion of a mass-spring system: $m = 1 \text{ kg}$, $k = 1(N/m)$, $x(0) = 2m$, r : damping constant.

FORCED HARMONIC OSCILLATIONS

Mechanical systems are often acted upon by an external force of large magnitude that acts for only a short period of time. For example a spring-mass system could be given a sharp blow at some specific time t . In engineering practice it is convenient to use the Dirac delta function as a mathematical model for such a blow. But it is my experience that it is often difficult for engineering students to get a real understanding of what this impulse function stands for. With the use of Maple it is easy and instructive to simulate Dirac's delta function $\delta(t - a)$.

The response of a mass-spring system to an impulse function

As said above the Dirac delta function $\delta(t - a)$ could serve as a mathematical model for an external force of large magnitude that acts for only a very short period of time.

```
>delta(t-a)=piecewise(t<a,0,t=a,infinity,t>a,0);
```

$$\delta(t - a) = \begin{cases} 0 & t < a \\ \infty & t = a \\ 0 & a < t \end{cases}$$

where:

```
>Int(delta(t-a),t=-infinity..infinity)=int(Dirac(t-a),t=-infinity..infinity);
```

$$\int_{-\infty}^{\infty} \delta(t - a) dt = 1$$

Obviously no real function can satisfy being zero except at a single point and have an integral equal to one. In Maple $\delta(t - a) = \text{Dirac}(t - a)$ is expressed as the derivative of the Heaviside standard unit step function $u(t - a)$.

```
>alias(u=Heaviside):
>Diff(u(t-a),t)=diff(u(t-a),t);
```

$$\frac{\partial}{\partial t} u(t - a) = \text{Dirac}(a - t)$$

It is instructive to use Maple to model such an instantaneous unit impulse by starting with the function:

```
>delta[epsilon](t-a)=piecewise(t<a,0,t<a+epsilon,1/epsilon,0);
```

$$\delta_{\epsilon}(t - a) = \begin{cases} 0 & t < a \\ \frac{1}{\epsilon} & t < a + \epsilon \\ 0 & \text{otherwise} \end{cases}$$

```
>h:=unapply(convert(rhs(%),Heaviside),t,a,epsilon);
```

$$h := (t, a, \epsilon) \rightarrow \frac{u(t - a)u(a + \epsilon - t)}{\epsilon}$$

A plot of the rectangular pulse for $a = \pi$ and $\epsilon = 3$ gives:

```
>plot(h(t,Pi,3),t=0..10,thickness=3);
>assume(epsilon>):
>A:=Int(h(t,a,epsilon),t=a..a+epsilon):
>A=value(A);
```

$$\int_a^{a+\epsilon} \frac{u(t - a)u(a + \epsilon - t)}{\epsilon} dt = 1$$




Fig. 10. Rectangular pulse $\delta_\epsilon(t - a)$ with $a = \pi$ and $\epsilon = 3$ (width ϵ and height $1/\epsilon$). The area of the rectangular pulse is equal to 1.

where $u(t)$ is Heavisides unit step function. We can compare the response of a damped mass-spring system to a rectangular pulse $\delta_\epsilon(t - a)$ as $\epsilon \rightarrow 0$ with the response of the Dirac's delta function used by Maple.

The output of the Maple code `impulse_func` in Fig. 11 animate the response to the rectangular pulse $\delta_\epsilon(t - a)$ by a linear second-order differential equation with constant coefficients and with damping.

```
>impulse_func(1,3,1,Pi,3,3,tickmarks=[0,0],labels=['', '']);
```

$$\left(\frac{\partial^2}{\partial t^2}y(t)\right) + 3\left(\frac{\partial}{\partial t}y(t)\right) + y(t) = \frac{5}{3}u(t - \pi) - \frac{5}{3}u(-\pi - 3 + t), \quad \epsilon = 3$$

$$\left(\frac{\partial^2}{\partial t^2}y(t)\right) + 3\left(\frac{\partial}{\partial t}y(t)\right) + y(t) = 5 \text{Dirac}(t - \pi)$$




Fig. 11. The behavior of the response to a rectangular pulse by a linear second-order differential equation with damping as $\epsilon \rightarrow 0$ compared to the response to Dirac's delta function.

The last animated frame in Fig. 11 shows little difference between the two responses when $\epsilon = 0.1$. Figure 12 shows an animated response of the damped mass-spring system initially at rest. At time $t = \pi$ the system is suddenly given a sharp “hammerblow” modelled by $f(t) = 5 \text{Dirac}(t - \pi)$.

```
>f:=t->5*Dirac(t-Pi):
>mass_spring(1,3,1,0,5,0,0,14,f,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + 3\left(\frac{\partial}{\partial t}x(t)\right) + x(t) = 5 \text{Dirac}(t - \pi)$$

$$x(t) = \sqrt{5}u(t - \pi)e^{(1/2(t-\pi)(-3+\sqrt{5}))} - \sqrt{5}u(t - \pi)e^{(-1/2(t-\pi)(3+\sqrt{5}))}$$




Fig. 12. The motion of a mass-spring system with dashpot under the influence of a sharp blow at $t = \pi$ provided by $5\delta(t - \pi)$.

The displacement resonance in forced oscillations

In every oscillating system there is dissipation of mechanical energy, which results when the motion of the mass-spring system described in the previous sections die out. If the oscillations are to be maintained, energy must be supplied to the system. In this section we shall assume that the system is acted on by a periodic driving force. Suppose that the mass-spring system is subjected to a periodic force $F_0 \sin(\omega t)$, where F_0 is the maximum value of the applied force and $f = \omega/2\pi$ is its frequency. The equation of motion is then:

$$m \frac{\partial^2}{\partial t^2}x(t) + r \frac{\partial}{\partial t}x(t) + kx(t) = F_0 \sin(\omega t)$$

where m is the mass of the system, $x(t)$ is the distance of the body of mass from its equilibrium position, r is the damping constant and k is the force constant. The frequency response of this system can be readily obtained from the system transfer function $H(s)$ by replacing s by $i\omega$.

```
>f:='f':
>with(inttrans): L:=x->laplace(x,t,s): invL:=X->invlaplace(X,s,t):
>alias(X(s)=L(x(t)),F(s)=L(f(t))):
>dign:=m*diff(x(t),t$2)+r*diff(x(t),t)+k*x(t)=f(t);
```

$$dign := m \left(\frac{\partial^2}{\partial t^2} x(t) \right) + r \left(\frac{\partial}{\partial t} x(t) \right) + kx(t) = f(t)$$

```
>simplify(L(dign));
```

$$ms^2X(s) - msx(0) - mD(x)(0) + rsX(s) - rx(0) + kX(s) = F(s)$$

```
>X(s)=solve(%,X(s));
```

$$X(s) = \frac{msx(0) + mD(x)(0) + rx(0) + F(s)}{ms^2 + rs + k}$$

```
>subs(x(0)=0,D(x)(0)=0,%);
```

$$X(s) = \frac{F(s)}{ms^2 + rs + k}$$

The transfer function of the mass-spring system is given by:

$$H(s) = \frac{1}{ms^2 + rs + k}$$

```
>H:=s->1/(m*s^2+r*s+k):
```

It is easy to show that the steady-state frequency response to an input $q(t) = F_0 \cos(\omega t)$ becomes:

```
>x(t)=F[0]*abs('H'(I*omega))*cos(omega*t+arg('H'(I*omega)));
```

$$x(t) = F_0 |H(I\omega)| \cos(\omega t + \arg(H(I\omega)))$$

```
>value(%);
```

$$x(t) = \frac{F_0 \cos\left(\omega t + \arg\left(\frac{1}{-m\omega^2 + I r \omega + k}\right)\right)}{|-m\omega^2 + I r \omega + k|}$$

The steady-state system response is also a cosine having the same frequency ω as the input. And the amplitude of this response is $F_0 |H(I\omega)|$. The variation in both the magnitude $|H(I\omega)|$ and argument $\arg(H(I\omega))$ as the frequency ω of the input cosine is varied constitute the frequency response of the system, as the following example shows.

Let us first solve the differential equation using Maples `dsolve`.

```
>f:=(omega,t)->2*cos(omega*t);
```

$$f := (\omega, t) \rightarrow 2 \cos(\omega t)$$

```
>deq:=(m,r,k,omega)->m*diff(x(t),t$2)+r*diff(x(t),t)+k*x(t)=f(omega,t);
```

$$deq := (m, r, k, \omega) \rightarrow m \left(\frac{\partial^2}{\partial t^2} x(t) \right) + r \left(\frac{\partial}{\partial t} x(t) \right) + kx(t) = f(\omega, t)$$

```
>sol:=dsolve({deq(1,0.25,4,2),x(0)=0,D(x)(0)=0},x(t),method=laplace);
```

$$sol := x(t) = 4. \sin(2.t) - 4.007835463e^{(-.1250000000t)} \sin(1.996089928t)$$

```
>plot(rhs(sol),t=0..40,color=black);
```




Fig. 13. The displacement $x(t)$ of a mass-spring system undergoing forced oscillations plotted against the time: $m = 1$ kg, $r = 0.25$ (Ns/m), $k = 4$ (N/m), $\omega = 2/s$.

The graph shows that the transient solution dies out as t increases.

The transfer function of the system is given by:

```
>H:=(m,r,k,s)->1/(m*s^2+r*s+k);
```

$$H := (m, r, k, s) \rightarrow \frac{1}{ms^2 + rs + k}$$

With $m = 1$ kg, $k = 4$ (N/m) and $s = I\omega$, we get:

```
>'H'(1,r,4,I*omega)=H(1,r,4,I*omega);
```

$$H(1, r, 4, I\omega) = \frac{1}{-\omega^2 + Ir\omega + 4}$$

```
>Habs:=unapply(evalc(abs(H(1,r,4,I*omega))),r,omega);
```

$$Habs := (r, \omega) \rightarrow \frac{1}{\sqrt{\omega^4 - 8\omega^2 + 16 + r^2\omega^2}}$$

There is a frequency called the resonance frequency at which the amplitude $A = 2Habs$ becomes a maximum. This resonance frequency can be recognized in many vibrating systems unless the damping force r is too large.

```
>solve(omega^4-8*omega^2+16+r^2*omega^2=0,{r});
```

$$\left\{ r = \frac{I(\omega + 2)(\omega - 2)}{\omega} \right\}, \quad \left\{ r = -\frac{I(\omega + 2)(\omega - 2)}{\omega} \right\}$$

With zero damping force, $r = 0$ and the resonance frequency $\omega = 2$ in our example. See also Fig. 15.

The phase angle ϕ is given by:

```
>phi:=unapply(evalc(argument(H(1,r,4,I*omega))),r,omega);
```

$$\phi := (r, \omega) \rightarrow \arctan(-r\omega, -\omega^2 + 4)$$

Substituting the values $r = 0.25(Ns/m)$, $\omega = 2/s$ and $F_0 = 2N$, gives the steady-state response, $x_s = x(t)$

```
>xs:=unapply(2*Habs(0.25,2)*cos(2*t+phi(0.25,2)),t);
'xs'(t)=xs(t); #steady state respons
```

$$x_s(t) = 4.000000000 \cos(2t - 1.570796327)$$

```
>sol; #Maple's solution
```

$$x(t) = 4. \sin(2.t) - 4.007835463e^{(-1.250000000t)} \sin(1.996089928t)$$

As $t \rightarrow \infty$, $x(t) \rightarrow x_s(t)$.

```
>control:='4*sin(2*t)-4*cos(2*t-Pi/2)'=simplify(4*sin(2*t)-4*cos(2*t-Pi/2));
```

$$control := 4 \sin(2t) - 4 \cos\left(2t - \frac{1}{2}\pi\right) = 0$$

The system is lagging because the phase shift between input and output $\arg(H(i\omega))$ is $\phi = -1.57 \dots$ Figure 14 shows this phase shift.

```
>with(plots):
plt1:=plot([f(2,t),xs(t)],t=30..39,color=black,thickness=[1,2]:
plttext1:=textplot([[39,xs(39),'steady-state
response'],[39,f(2,39),'input']],align=RIGHT):
display({plt1,plttext1},labels=['t','x(t)'],labelfont=[TIMES,BOLD,12]);
```




Fig. 14. Phase shift between steady-state response and input displacement $x(t)$.

```
>A:='A':
Habsi:=i->plot(Habs(i,omega),omega=1..3,A=0..2.5,color=black):
>plt2:=i->display(Habsi(i),text2(i)):
>text2:=i->textplot([2,Habs(i,2),convert(r=i,string)],align=ABOVE,font=[TIMES,BOLD,12]:
>display(seq(plt2(i),i=[0.1,0.25,0.5,0.5,0.75,1,1.41]),labels=['omega','A'],labelfont=[TIMES,BOLD,12]);
```

Fig. 15. Variation of the amplitude $A = 2/(\sqrt{\omega^4 - 8\omega^2 + 16 + r^2\omega^2})$ against the angular velocity ω of the forced oscillations-displacement resonance for different values of the damping constant r .

Figure 15 shows that the smaller the value of r , the sharper the resonance of the vibrating system to the applied force. And the resonance frequency varies with the value of the damping constant r . The amplitude A tends towards infinity when the applied frequency becomes equal to the resonance frequency and the system has zero damping.




Fig. 15.

Figure 16 animates the displacement resonance of the forced oscillations and Fig. 17 shows the variation of the phase shift against the angular velocity, which is animated in Fig. 18. Figure 19 shows what happens with the mass-spring system when the damping constant $r = 0.25(N/s)$ and Fig. 20 when the damping constant is approximately equal to zero, $r = 0.01(N/s)$.

```
>display(seq(plt2(0.05*i),i=1..28),insequence=true,labels=['omega','A'],labelfont=[TIMES,BOLD,12]);
```




Fig. 16. Animation of Figure 15.


```
>Harg:=i->plot(phi(i,omega),omega=1..3,color=black):
>text3:=i->textplot([2.4,phi(i,2.4),convert(r=i,string)],align={RIGHT,ABOVE},font=[TIMES,BOLD,12]):
>plt3:=i->display(Harg(i),text3(i)):
>display(seq(plt3(i),i=[0.1,0.25,0.5,0.75,1,1.41]),labels=['omega','phi'],labelfont=[TIMES,BOLD,12]);
```




Fig. 17. Variation of the phase shift $\phi = \arctan(\omega^2 - 4/r\omega)$ against the angular velocity ω of the forced oscillations-displacement resonance for different values of the damping constant r .

As the applied velocity is increased from zero to the natural angular velocity of the system, the phase angle ϕ increases from zero to $\pi/2$. When $r = 0$ the angle ϕ changes abruptly from 0 to π .

```
>display(seq(plt3(0.05*i),i=1..28),labelfont=[TIMES,BOLD,12],insequence=true);
```




Fig. 18. Animation of Fig. 17.

```
>f:=t->2*sin(2*t):
>mass_spring(1,0.25,4,0,5,0,0,20,f,3,tickmarks=[5,3],labels=['t',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + 0.25\left(\frac{\partial}{\partial t}x(t)\right) + 4x(t) = 2\sin(2t)$$

$$x(t) = -4.000000000\cos(2.000000000t) + 530\ 10^{-8}\sin(2.000000000t)$$

$$+ 4e^{(-0.1250000000t)}\cos(1.996089928t)$$

$$+ 0.2504897111e^{(-0.1250000000t)}\sin(1.996089928t)$$




Fig. 19. The displacement $x(t)$ of the mass-spring system undergoing forced oscillations with damping constant $r = 0.25$.

```
>mass_spring(1,0.01,4,0,5,0,0,11.5,f,3,tickmarks=[5,3],labels=['',''],scaling=constrained);
```

$$\left(\frac{\partial^2}{\partial t^2}x(t)\right) + .01\left(\frac{\partial}{\partial t}x(t)\right) + 4x(t) = 2\sin(2t)$$

$$x(t) = -99.99999997\cos(2.000000000t) + 0.19530\ 10^{-6}\sin(2.000000000t)$$

$$+ 99.99999995e^{(-0.005000000000t)}\cos(1.999993750t)$$

$$+ 0.2500005858e^{(-0.005000000000t)}\sin(1.999993750t)$$

Fig. 20. The displacement $x(t)$ of the mass-spring system undergoing forced oscillations with damping constant $r = 0.01(Ns/m)$.




Fig. 20.

SUMMARY

This paper discusses the use of the computer algebra system (CAS) Maple in calculating and animating the responses of free and force vibrations systems. CAS has changed the fundamental way in which many math and engineering courses are taught. The user can enter a mathematical expression which in this paper Maple subsequently executes, and one is freed from tedious computations. This allows the student to explore how the solution of an engineering problem depends on various parameters of the problem under study. The graphics capabilities are extremely helpful for visualizing the behaviour of the system under investigation. Maple's plots and plottools packages enable the user to utilize plotting objects for animating purposes. This is highlighted and demonstrated in this paper. It is hoped that this feature can attract the attention of both lecturers and students to visualize dynamic responses, typical in engineering, more realistically.

Harald Pleym is Associate Professor in Department of Technology at Telemark College in Porsgrunn, Norway. His professional background is geophysics and atmospheric sciences, but now he regularly teaches courses in engineering mathematics. He initiated the use of the Computer Algebraic System, Maple, as a tool in all math courses since August 1995. His current main interest is preparing lessons and developing supplementary, interactive course material in Maple. The Maple system enables him to develop both simple and complex mathematical models in the classroom and laboratory, making mathematics more dynamic and meaningful.